


Device Modeling Report

COMPONENTS : VOLTAGE COMPARATOR (CMOS)
PART NUMBER : NJU7108
MANUFACTURER : NEW JAPAN RADIO



Bee Technologies Inc.

Spice Model



```

*$
*PART NUMBER: NJU7108
*MANUFACTURER: NEW JAPAN RADIO
*Push Pull
*All Rights Reserved Copyright (c) Bee Technologies Inc. 2005
.subckt nju7108 In- VSS In+ OUT VDD
I1          1 VSS DC 24u
I2          VSS In+ DC 1.5p
I3          VSS In- DC 0.5p
M1          3 In- 4 VDD MbreakPD3
M2          3 In+ 5 VDD MbreakPD2
M3          1 1 VDD VDD MbreakPD
M4          3 1 VDD VDD MbreakPD
M5          2 2 VDD VDD MbreakPD
M6          2 4 VSS VSS MbreakND1
M7          4 4 VSS VSS MbreakND1
M8          5 4 VSS VSS MbreakND1
M9          6 2 VDD VDD MbreakPD
M10         7 6 VDD VDD MbreakPD
M11         6 5 VSS VSS MbreakND
M12         7 6 VSS VSS MbreakND
M13         OUT 7 VDD VDD MbreakPD1
M14         OUT 7 VSS VSS MbreakND2
C1          OUT 6 6.3p
C2          OUT 7 6.3p
.model MbreakND NMOS (LEVEL=3 L=0.01u W=1m VTO=0.9 RS=10.000E-3
+ RD=10.000E-3 RDS=1.0000E6 TOX=2.0000E-6 RG=5 RB=1.0000E-3 KP=1E-6)
.model MbreakND1 NMOS (LEVEL=3 L=27u W=0.1 VTO=1 RS=10.000E-3
+ RD=10.000E-3 RDS=1.0000E6 TOX=2.0000E-6 RG=5 RB=1.0000E-3 KP=1E-6)

```


```
.model MbreakND2 NMOS (LEVEL=3 L=3u W=2.6m VTO=0.9 RS=10.000E-3
+ RD=10.00E-3 RDS=1.0000E6 TOX=2.0000E-6 RG=5 RB=1.0000E3 KP=10.E-6)
.model MbreakPD PMOS (LEVEL=3 L=0.5u W=0.02 VTO=-1 RS=10.000E-3
+ RD=10.000E-3 RDS=1.0000E6 TOX=2.0000E-6 RG=5 RB=1.0000E-3 KP=1E-6)
.MODEL MbreakPD1 PMOS (LEVEL=3 L=6u W=.575 VTO=-0.9 RS=10.000E-3
+ RD=10.000E-3 RDS=1.0E6 TOX=2.0000E-6 RG=5 RB=1.0000E-3 KP=0.09E-6)
.MODEL MbreakPD2 PMOS (LEVEL=3 L=6u W=0.1 VTO=-1.5 RS=10.000E-3
+ RD=10.00E-3 RDS=1.2900E6 TOX=2.0000E-6 RG=5 RB=1.0000E-3 KP=1E-6)
.MODEL MbreakPD3 PMOS (LEVEL=3 L=6u W=0.0835 VTO=-1.5 RS=10.000E-3
+ RD=10.00E-3 RDS=1.14800E6 TOX=2.0000E-6 RG=5 RB=1.0000E-3 KP=1E-6)
.ENDS
*$
```

MODEL PARAMETER


Pspice model parameter	Model description
LEVEL	
L	Channel Length
W	Channel Width
KP	Transconductance
RS	Source Ohmic Resistance
RD	Ohmic Drain Resistance
VTO	Zero-bias Threshold Voltage
RDS	Drain-Source Shunt Resistance
TOX	Gate Oxide Thickness
CGSO	Zero-bias Gate-Source Capacitance
CGDO	Zero-bias Gate-Drain Capacitance
CBD	Zero-bias Bulk-Drain Junction Capacitance
MJ	Bulk Junction Grading Coefficient
PB	Bulk Junction Potential
FC	Bulk Junction Forward-bias Capacitance Coefficient
RG	Gate Ohmic Resistance
IS	Bulk Junction Saturation Current
N	Bulk Junction Emission Coefficient
RB	Bulk Series Resistance
PHI	Surface Inversion Potential
GAMMA	Body-effect Parameter
DELTA	Width effect on Threshold Voltage
ETA	Static Feedback on Threshold Voltage
THETA	Modility Modulation
KAPPA	Saturation Field Factor
VMAX	Maximum Drift Velocity of Carriers
XJ	Metallurgical Junction Depth
UO	Surface Mobility

Output Low Voltage

Simulation result



Evaluation Circuit




Comparison Table


$I_{OL}=5mA$	Measurement	Simulation	%Error
$V_{OL} (V)$	0.3	0.295787	-1.404

Output High Voltage

Simulation result



Evaluation Circuit




Comparison Table


$I_{OH} = -5mA$	Measurement	Simulation	%Error
V_{OH} (V)	2.7	2.7027	0.1

Input Bias Current and Input Offset Current Characteristics

Simulation result



Evaluation Circuit




Comparison Table


	Measurement	Simulation	%Error
I_b (pA)	1	0.978286	-2.171
I_{os} (pA)	1	0.972171	-2.783

Input Offset Voltage Characteristics

Simulation result



Evaluation Circuit




Comparison Table


	Measurement	Simulation	%Error
V_{Io} (mV)	4	3.9525	-1.188

Propagation Delay Time and Response Time

Simulation result



Evaluation Circuit



Comparison Table

Over drive=100mV	Measurement	Simulation	%Error
t_{PLH} (ns)	500	498.479	-0.304
t_{PHL} (ns)	190	190.649	0.342
t_{TLH} (ns)	10	9.694	-3.060
t_{THL} (ns)	5	5.1555	3.110