


# Device Modeling Report

COMPONENTS : C-MOS COMPARATOR  
PART NUMBER : NJU7116  
MANUFACTURER : NEW JAPAN RADIO



**Bee Technologies Inc.**

## Spice Model



```

*$
*PART NUMBER: NJU7116
*MANUFACTURER: NEW JAPAN RADIO
*Push Pull
*All Rights Reserved Copyright (c) Bee Technologies Inc. 2006
.subckt nju7116 In- VSS In+ OUT VDD
I1          1 VSS DC 20u
I2          VSS In+ DC 1.5p
I3          VSS In- DC 0.5p
M1          3 In- 4 VDD MbreakPD3
M2          3 In+ 5 VDD MbreakPD2
M3          1 1 VDD VDD MbreakPD
M4          3 1 VDD VDD MbreakPD
M5          2 2 VDD VDD MbreakPD
M6          2 4 VSS VSS MbreakND1
M7          4 4 VSS VSS MbreakND1
M8          5 4 VSS VSS MbreakND1
M9          6 2 VDD VDD MbreakPD
M10         7 6 VDD VDD MbreakPD
M11         6 5 VSS VSS MbreakND
M12         7 6 VSS VSS MbreakND
M13         OUT 7 VDD VDD MbreakPD1
M14         OUT 7 VSS VSS MbreakND2
C1          6 2 10p
C2          2 7 1.5p
.model MbreakND NMOS (LEVEL=3 L=6u W=8m VTO=0.9 RS=10.000E-3
+ RD=10.000E RDS=1.0000E6 TOX=2.0000E-6 CBD=1.90000e-11
+ RG=5 RB=1.0000E-3 KP=1E-6)

```

```

.model MbreakND1 NMOS (LEVEL=3 L=27u W=0.0036 VTO=1
+ RS=10.000E-3 RD=10.000E-3 RDS=1.0000E6 TOX=2.0000E-6
+ CBD=16.000e-12 RG=5 RB=1.0000E-3 KP=1E-6)
.model MbreakND2 NMOS (LEVEL=3 L=3u W=1.05m VTO=0.9
+ RS=10.000E-3 RD=10.000E-3 RDS=1.0000E6 TOX=2.0000E-6
+ CBD=1.0000e-12 CGSO=1.00000e-9 CGDO=1.0000e-12 RG=5
+ RB=1.0000E3 KP=10.E-6)
.model MbreakPD PMOS (LEVEL=3 L=0.5u W=0.004 VTO=-1 RS=10.000E-3
+ RD=10.000E-3 RDS=1.0000E6 TOX=2.0000E-6 RG=5
+ RB=1.0000E-3 KP=1E-6)
.MODEL MbreakPD1 PMOS (LEVEL=3 L=6u W=.3 VTO=-0.9
+ RS=10.000E-3 RD=10.000E-3 RDS=1.0E6 TOX=2.0000E-6
+ RG=5 RB=1.0000E-3 KP=0.09E-6)
.MODEL MbreakPD2 PMOS (LEVEL=3 L=6u W=0.1 VTO=-1.5 RS=10.000E-3
+ RD=10.00E-3 RDS=1.2900E6 TOX=2.0000E-6 RG=5
+ RB=1.0000E-3 KP=1E-6)
.MODEL MbreakPD3 PMOS (LEVEL=3 L=6u W=0.0925 VTO=-1.5
+ RS=10.000E-3 RD=10.00E-3 RDS=1.0401100E6 TOX=2.0000E-6
+ RG=5 RB=1.0000E-3 KP=1E-6)
.ENDS
*$


```

## MODEL PARAMETER


Pspice model parameter	Model description
LEVEL	
L	Channel Length
W	Channel Width
KP	Transconductance
RS	Source Ohmic Resistance
RD	Ohmic Drain Resistance
VTO	Zero-bias Threshold Voltage
RDS	Drain-Source Shunt Resistance
TOX	Gate Oxide Thickness
CGSO	Zero-bias Gate-Source Capacitance
CGDO	Zero-bias Gate-Drain Capacitance
CBD	Zero-bias Bulk-Drain Junction Capacitance
MJ	Bulk Junction Grading Coefficient
PB	Bulk Junction Potential
FC	Bulk Junction Forward-bias Capacitance Coefficient
RG	Gate Ohmic Resistance
IS	Bulk Junction Saturation Current
N	Bulk Junction Emission Coefficient
RB	Bulk Series Resistance
PHI	Surface Inversion Potential
GAMMA	Body-effect Parameter
DELTA	Width effect on Threshold Voltage
ETA	Static Feedback on Threshold Voltage
THETA	Modility Modulation
KAPPA	Saturation Field Factor
VMAX	Maximum Drift Velocity of Carriers
XJ	Metallurgical Junction Depth
UO	Surface Mobility

## Output Low Voltage

### Simulation result



### Evaluation Circuit




### Comparison Table


$I_{OL}=2mA$	Measurement	Simulation	%Error
$V_{OL} (V)$	0.3	0.292932	-2.35600

## Output High Voltage

### Simulation result



### Evaluation Circuit




### Comparison Table


$I_{OH} = -2mA$	Measurement	Simulation	%Error
$V_{OH} (V)$	2.7	2.7760	2.81481

# Input Bias Current and Input Offset Current Characteristics

## Simulation result



## Evaluation Circuit




## Comparison Table


	Measurement	Simulation	%Error
$I_b$ (pA)	1	1	0
$I_{os}$ (pA)	1	1	0

# Input Offset Voltage Characteristics

Simulation result



Evaluation Circuit




Comparison Table

	Measurement	Simulation	%Error
V <sub>IO</sub> (mV)	2.5	2.4500	-2.00000




# Common-Mode Rejection Ratio

## Simulation result



## Evaluation Circuit




$$CMRR = 20\log \left( \left| \frac{V_{out1} - V_{out2}}{V_{in1} - V_{in2}} \right| \times \frac{R_s}{R_f + R_s} \right)$$

## Comparison Table


	Measurement	Simulation	%Error
<b>CMRR (dB)</b>	50	50.897	1.79400

## Propagation Delay Time and Response Time

### Simulation result



### Evaluation Circuit



### Comparison Table

Over drive=100mV	Measurement	Simulation	%Error
$t_{PLH}$ (us)	1.2	1.2195	1.62500
$t_{PHL}$ (us)	3.3	3.3291	0.88182
$t_{TLH}$ (ns)	15	14.554	-2.97333
$t_{THL}$ (ns)	40	40.782	1.95500